
Capítulo 1 

¿Por qué tu regresión miente? Correlación y 
causalidad 

Hace unas décadas, los epidemiólogos tenían una pista prometedora contra el cáncer de pulmón. Las personas 
que comían muchas frutas y verduras ricas en carotenoides, y las que tenían niveles altos de betacaroteno en la 
sangre, enfermaban menos. Los estudios lo repetían una y otra vez. 

¿Qué más hacía falta? Si el betacaroteno protege, basta con dárselo en una pastilla a quienes más lo 
necesitan: los fumadores. 

En el suroeste de Finlandia alguien decidió comprobarlo como se debe. Se reclutó a 29 133 fumadores 
varones de 50 a 69 años. No se dejó que cada uno eligiera qué tomar. Se sorteó: unos recibieron vitamina E, 
otros betacaroteno, otros ambas cosas y otros una pastilla sin nada activo. Ni ellos ni quienes los atendían sabían 
qué le había tocado a cada uno. Los siguieron durante cinco a ocho años. 

En 1994 se publicaron los resultados. Entre los hombres que recibieron betacaroteno hubo más cáncer de 
pulmón, no menos: un 18 % más que entre los que no lo recibieron. Su mortalidad total también fue mayor, en 
un 8 %. Los propios autores lo calificaron de inesperado (Alpha-Tocopherol, Beta Carotene Cancer Prevention 
Study Group, 1994). 

Años de estudios decían «protege». El experimento dijo «no protege, y quizá daña». ¿Cómo pudo 
equivocarse tanta evidencia a la vez? 

Piensa en quién tiene el betacaroteno alto en la sangre. Por lo general, alguien que come bien. Y quien come 
bien suele diferir en muchas otras cosas: cuánto fuma, cuánto se mueve, cuánto gana, cuán a menudo va al 
médico. Los estudios observacionales comparaban a personas distintas y le atribuían toda la diferencia a una sola 
molécula. El sorteo, en cambio, comparó a hombres iguales en todo, salvo en la pastilla. 

Ahora piensa en tu investigación. Tienes una encuesta, una variable que te interesa y un programa que calcula 
una regresión en dos segundos. Te sale un coeficiente positivo con tres estrellas. ¿Cómo sabes que no te está 
pasando lo mismo? 

De eso trata este capítulo. Y, en el fondo, de eso trata todo el libro. 

Lo que vas a lograr 

Al terminar este capítulo podrás: 

•  Distinguir cuándo un número describe una relación y cuándo mide un efecto. 

•  Explicar con una sola ecuación por qué una comparación simple puede engañarte. 

•  Simular en EViews, paso a paso, un caso donde la regresión miente y otro donde dice la verdad. 



Capítulo 1   ¿Por qué tu regresión miente? Correlación y causalidad    3 

 

•  Reconocer en tu propia tesis la pregunta que el jurado te va a hacer primero. 

1.1   La correlación no te dice por qué 

Empecemos por lo que ya sabes. Dos variables están correlacionadas cuando se mueven juntas: si una sube, la 
otra tiende a subir (o a bajar). La correlación es útil, porque te dice que hay algo. Sin embargo, no te dice qué 
hay. 

Franz Messerli, un cardiólogo de Nueva York, lo mostró con un ejemplo que dio la vuelta al mundo. Reunió 
datos de 23 países y comparó cuánto chocolate se consume por habitante con cuántos premios Nobel ha ganado 
cada país por habitante. La relación salió fuerte y muy significativa, con Suiza a la cabeza de ambas listas. 
Messerli publicó el trabajo en una de las revistas médicas más leídas del mundo, y su propósito era advertir lo 
fácil que es tomarse en serio una correlación que no tiene sentido (Messerli, 2012). 

¿Por qué pasa algo así? Cuando dos variables se mueven juntas, hay cuatro explicaciones posibles. Conviene 
que las tengas a mano, porque el jurado las va a tener. 

1. Causalidad. X provoca Y. Por ejemplo, más fertilizante sube el rendimiento de una parcela, y una helada 
reduce la cosecha de papa. 

2. Causalidad inversa. Y provoca X. Imagina que encuentras que las empresas que más gastan en publicidad 
son las que más venden. ¿La publicidad genera las ventas? Puede ser, pero muchas empresas fijan su presupuesto 
de publicidad como una parte de lo que venden. Entonces vender más hace gastar más en publicidad, y no solo al 
revés. 

3. Una tercera variable. Algo distinto mueve a las dos. En el caso del chocolate, una candidata obvia es la 
riqueza del país: un país rico consume más chocolate y también financia más universidades y laboratorios. En la 
historia del betacaroteno, comer bien iba de la mano con muchos otros hábitos saludables. A esa tercera variable 
los econometristas le dicen variable de confusión o factor de confusión. 

4. El azar. Por ejemplo, las exportaciones de quinua y las lluvias de una ciudad lejana pueden moverse juntas 
durante algunos años sin tener nada que ver. Con suficientes variables, algunas se van a correlacionar por pura 
casualidad. Si cruzas cien pares de variables sin ninguna relación, unas cinco te saldrán «significativas» al 5 %. 
Así funciona ese umbral. 

Fíjate en algo incómodo. Las cuatro explicaciones producen la misma tabla en tu programa estadístico. Un 
coeficiente, un error estándar, unas estrellas. El software no sabe cuál de las cuatro es la verdadera. Tú sí tienes 
que saberlo. 

1.2   La pregunta que de verdad importa: ¿qué habría pasado si…? 

Vamos a un caso más cercano. Supongamos que Diego prepara su tesis sobre el rendimiento escolar en Trujillo. 
Tiene datos de 5 000 estudiantes de quinto de secundaria: sus notas finales, en la escala de 0 a 20, y si pagaron o 
no clases particulares. Los que pagaron clases sacaron, en promedio, tres puntos más. ¿Las clases particulares 
suben la nota en tres puntos? 

La pregunta que Diego quiere responder no es «¿quiénes sacaron más nota?». Es otra, más difícil: ¿cuánto 
habría sacado cada estudiante si no hubiera tomado clases? 

Esa nota que no vemos tiene nombre. Se llama resultado potencial. Cada estudiante tiene dos: 

•  La nota que sacaría con clases particulares. 

•  La nota que sacaría sin clases particulares. 

El efecto de las clases para ese estudiante es la diferencia entre las dos. Simple. El problema es que la vida no 
nos deja ver las dos. Cada estudiante tomó clases o no las tomó. Uno de sus dos resultados potenciales quedó 
para siempre en el terreno del «¿qué habría pasado si…?». Los libros lo llaman el problema fundamental de la 
inferencia causal (Cunningham, 2021). 

¿Entonces no se puede medir nada? Sí se puede, aunque no estudiante por estudiante: se mide en promedio, y 
con cuidado. Y el cuidado empieza por entender dónde está la trampa. 

1.3   Dibuja tu problema antes de estimarlo 

Antes de abrir EViews, toma una hoja y dibuja flechas. Parece un juego de niños, pero es una de las herramientas 
más serias de la inferencia causal moderna. Cada flecha significa «esto provoca aquello». El problema de Diego, 
dibujado, se ve así: 
   Motivación ──────────► Nota 
       │                   ▲ 
       ▼                   │ 
    Clases ────────────────┘ 

La flecha de abajo, de Clases a Nota, es la que Diego quiere medir. Las otras dos son el problema: la 
motivación empuja al estudiante hacia las clases y, por otro camino, empuja también su nota hacia arriba. Ese 
camino alternativo, que pasa por la motivación, se llama camino de puerta trasera. Mientras esté abierto, parte 
de la relación entre clases y nota viaja por él, y tu regresión la confunde con el efecto de las clases (Cunningham, 
2021). 

Dibujar te obliga a hacer explícito lo que crees sobre el mundo. Si alguien te dice «te faltó considerar el 
ingreso familiar», no discuten sobre estrellas y valores p: discuten sobre una flecha. ¿El ingreso familiar mueve 
la probabilidad de pagar clases? Seguro que sí. ¿Mueve la nota por otro camino? Probablemente, porque un 
hogar con más ingresos también compra libros, internet y tranquilidad. Entonces es otra puerta trasera que tienes 
que cerrar. 

Hay tres maneras de cerrar una puerta trasera, y este libro las recorre todas. Puedes medir el confusor y 
controlarlo, usar datos donde el confusor no cambia o encontrar una fuente de variación que no pase por él. Por 
ejemplo, un sorteo. Por ahora basta con que dibujes. Vas a ver que la mitad de las preguntas del jurado se 
responden mirando ese dibujo. 

1.4   La trampa, en una sola línea 

Cuando Diego compara el promedio de los que tomaron clases con el de los que no, está calculando esto: 

Diferencia observada = efecto causal + sesgo de selección 

Esta ecuación es la más importante del libro. Angrist y Pischke la ponen en el centro de su manual, y con 
razón (Angrist & Pischke, 2014). Léela despacio. 

El efecto causal es lo que Diego quiere: cuánto suben la nota las clases a quienes las tomaron. 

El sesgo de selección es la diferencia que ya existía entre los dos grupos antes de las clases. ¿Quiénes pagan 
clases particulares? Los estudiantes más motivados. Los que tienen padres pendientes de sus notas. Los que 
tienen algo de dinero extra en casa. Esos estudiantes habrían sacado mejor nota aunque no hubieran tomado ni 
una clase. 

Si el sesgo de selección es grande, la diferencia observada puede ser el triple del efecto real. O puede tener el 
signo contrario, como en la historia del betacaroteno. Y ningún valor p te va a avisar. 

Por eso los econometristas hacemos siempre la misma pregunta incómoda: ¿los dos grupos eran 
comparables antes del tratamiento? Si la respuesta es no, tu diferencia mezcla dos cosas y no sabes cuánto 
pesa cada una. 

1.5   El truco del experimento 

¿Cómo se elimina el sesgo de selección? Con una moneda. 

Si Diego pudiera sortear quién recibe clases particulares, la motivación, el dinero y los padres atentos 
quedarían repartidos más o menos igual entre los dos grupos. Nadie eligió. El azar eligió. Entonces el sesgo de 
selección se va a cero y la diferencia observada mide el efecto causal. Eso hizo el estudio de Finlandia, y por eso 
encontró lo que años de estudios observacionales no vieron. 

América Latina tiene su propio ejemplo famoso. 

Colombia tuvo un programa, PACES, que entregaba vouchers para pagar parte de la pensión de un colegio 
secundario privado. Los vouchers se renovaban solo si el estudiante mantenía un rendimiento satisfactorio. Y los 
vouchers se repartieron por sorteo (Angrist et al., 2002). 

Como el azar decidió quién ganaba, los ganadores y los perdedores eran comparables, y la diferencia entre 
ellos se podía leer como efecto del voucher. Tres años después de los sorteos, los ganadores tenían unos 10 
puntos porcentuales más de probabilidad de haber terminado el 8.º grado, sobre todo porque repetían menos de 
año. Además, sacaban 0,2 desviaciones estándar más en las pruebas de rendimiento (Angrist et al., 2002). 

Todo empezó con un sorteo. 

1.6   ¿Y si no puedes hacer un experimento? 

Aquí viene la mala noticia: casi nunca vas a poder. No puedes sortear quién estudia una maestría, quién vive en 
Lima o quién recibe un crédito. Tu investigación probablemente usa una encuesta que ya existe, como la 
ENAHO, o datos de una empresa, o series del Banco Central. 

La buena noticia es que la econometría es, justamente, el arte de acercarse al experimento cuando no lo 
tienes. Este libro recorre las herramientas en orden de dificultad: 

•  Controlar por lo que sí observas (capítulos 4 a 8). Si el confusor está en tus datos, lo metes en la 
regresión. 

•  Seguir a las mismas personas o empresas en el tiempo (capítulo 9). Los datos de panel eliminan lo que 
no cambia, aunque no lo midas. 

•  Buscar algo que se parezca al azar (capítulos 12 a 14). Variables instrumentales, diferencias en 
diferencias, regresión discontinua y emparejamiento. 

Cada herramienta tiene un supuesto. Tu trabajo, y el de este libro, es saber cuál es y defender por qué se 
cumple en tu caso. 

Basta de teoría. Vamos a ver la trampa con tus propios ojos. 

1.7   Manos a la obra 

Abre EViews. En la parte superior verás la ventana Command: ahí escribes una instrucción, pulsas Enter y 
EViews la ejecuta. Copia los bloques en orden, línea por línea. Si prefieres tenerlos todos juntos, abre el archivo 

cap01.prg del repositorio del libro con File › Open › Programs y pulsa el botón Run; el prefacio te explica 
cómo descargarlo. No necesitas instalar nada adicional: todo lo que usamos viene con EViews. 

Vamos a hacer algo que en la vida real es imposible: crear un mundo donde conocemos la verdad. 
Simularemos a los 5 000 estudiantes de Diego. Decidiremos nosotros que las clases particulares suben la nota 
exactamente 1 punto. Y luego veremos qué nos dice cada método. Los datos de esta sección son simulados, no 
provienen de ninguna encuesta. 

La clave está en una variable que nadie mide en las encuestas: la motivación. En nuestro mundo, la 
motivación hace dos cosas. Sube la nota. Y aumenta la probabilidad de pagar clases. 

Paso 1. Crea 5 000 estudiantes con una motivación oculta 
wfcreate(wf=cap01) u 5000 
rndseed 2026 
series motivacion = nrnd 
show motivacion.hist 

 
FIGURA 1.1   Salida de EViews: motivación simulada de 5 000 estudiantes, con su histograma y sus estadísticas. Fuente: 
elaboración propia con EViews 14 y datos simulados (cap01.prg). 

¿Qué hicimos? wfcreate crea un workfile, el archivo de trabajo de EViews: wf=cap01 le pone nombre, u 
indica que los datos no tienen fechas (son personas, no meses) y 5000 crea 5 000 filas, una por estudiante. 

rndseed 2026 fija la semilla del generador de números aleatorios: gracias a ella, tú obtendrás exactamente 
los mismos números que aparecen en el libro. La función nrnd le da a cada estudiante una motivación al azar, 
con promedio 0 y desviación estándar 1. Por último, show motivacion.hist abre la serie en su vista de 
histograma. 

La ventana confirma lo que pedimos: una media de −0,009 y una desviación estándar de 1,008. Nadie mide 
esta motivación en una encuesta, pero en nuestro mundo existe. 

Un consejo práctico: corre los bloques una sola vez y en orden. Si repites un bloque que sortea números, la 
secuencia cambia y tus cifras dejarán de coincidir con las del libro. Si te pasa, vuelve a empezar desde el paso 1. 

Paso 2. Haz que los más motivados paguen clases 
series clases = @rbinom(1, @logit(1.5*motivacion)) 
show clases.freq 

 
FIGURA 1.2   Salida de EViews: estudiantes con y sin clases particulares en la simulación. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

Esta línea es el corazón del problema. La función @logit() convierte la motivación en una probabilidad entre 0 
y 1. Por ejemplo, un estudiante con motivación 0 tiene 50 % de probabilidad de pagar clases, y uno muy 
motivado se acerca al 100 %. Luego @rbinom(1, p) lanza una moneda cargada con esa probabilidad y anota 1 
si el estudiante pagó clases y 0 si no. La vista freq cuenta cuántos hay de cada tipo: 2 496 estudiantes (49,92 %) 
tomaron clases y 2 504 no. 

Paso 3. Construye la nota con un efecto real de 1 punto 
series ruido = 1.5*nrnd 
series nota = 11 + 1*clases + 2*motivacion + ruido 
nota = @recode(nota<0, 0, @recode(nota>20, 20, nota)) 
group g3 nota motivacion ruido 
show g3.stats 

FIGURA 1.3   Salida de EViews: estadísticas descriptivas de la nota, la motivación y el ruido. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

Lee la segunda línea despacio, porque ahí está la verdad que luego intentaremos descubrir. La nota parte de 11 
puntos. Suma 1 punto si el estudiante tomó clases, 2 puntos por cada unidad de motivación y un poco de ruido, 
que representa la suerte del día del examen. La tercera línea recorta la nota para que no baje de 0 ni pase de 20: 
@recode(condición, a, b) devuelve a si la condición se cumple y b si no. 

Las dos últimas líneas presentan un objeto nuevo. Un group junta varias series para verlas lado a lado, y su 
vista stats calcula las estadísticas de todas a la vez. El resumen muestra una nota promedio de 11,47 y una 
motivación promedio prácticamente nula (−0,009). 

Paso 4. Primer intento: comparar promedios 

Esto es lo que haría la mayoría. Separar en dos grupos y comparar. 
show nota.statby clases 

FIGURA 1.4   Salida de EViews: nota promedio de quienes tomaron clases y de quienes no. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

La vista statby calcula las estadísticas de una serie por cada valor de otra. Los que no tomaron clases 
promedian 9,89; los que sí, 13,05. La diferencia es de 3,15 puntos. Sabemos que la verdad es uno. La 
comparación simple triplicó el efecto. 

Un detalle de lectura: EViews escribe los decimales con punto, como en inglés. En el texto del libro usamos 
la coma, así que 13.04540 en la pantalla es 13,05 en la prosa. 

Paso 5. Segundo intento: la regresión 

«Bueno», dirás, «pero yo no comparo promedios, yo corro una regresión». Veamos. 
equation eq05.ls nota c clases 
show eq05 

FIGURA 1.5   Salida de EViews: regresión de la nota sobre las clases particulares, sin controles. Fuente: elaboración propia 
con EViews 14 y datos simulados (cap01.prg). 

La primera línea crea un objeto equation llamado eq05 y lo estima por mínimos cuadrados (ls). Después del 
punto va la variable dependiente y luego las explicativas; la c es la constante. EViews no la agrega solo: si la 
olvidas, estimas una recta que pasa por el origen. 

El coeficiente de CLASES es 3,15. El valor p (columna Prob.) es 0,0000. ¿Y el intervalo de confianza? 
EViews no lo muestra en esta tabla, pero lo tiene a un comando de distancia. 
eq05.cinterval 0.95 

FIGURA 1.6   Salida de EViews: intervalos de confianza al 95 % de la regresión sin controles. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

El intervalo de confianza al 95 % va de 3,02 a 3,28: ni siquiera se acerca al 1 verdadero. Es la tabla más 
convincente que verás en tu vida, y está equivocada. 

Mira además la constante, C: 9,89. Es exactamente el promedio de los que no tomaron clases en el paso 4. Y 
9,89 más 3,15 da 13,05: el promedio de los que sí las tomaron. 

Detente aquí un momento, porque esta es la lección del capítulo. Una regresión con una sola variable es 
exactamente la misma comparación de promedios, solo que con más decimales. Ni el valor p ni el tamaño de 
la muestra corrigen el sesgo de selección. Con 50 000 estudiantes, el error sería igual de grande; solo estarías 
más seguro de él. 

Paso 6. Tercer intento: si pudieras medir la motivación 

En nuestro mundo simulado tenemos algo que Diego no tiene: la motivación de cada estudiante. Metámosla en la 
regresión. 
equation eq06.ls nota c clases motivacion 
show eq06 



12    Parte Uno   Antes de estimar 

 

 
FIGURA 1.7   Salida de EViews: regresión de la nota sobre las clases, controlando por la motivación. Fuente: elaboración 
propia con EViews 14 y datos simulados (cap01.prg). 

Ahora el coeficiente de CLASES es 0,97, con un error estándar de 0,05: el 1 verdadero queda a menos de un error 
estándar. La regresión dejó de mentir. ¿Qué cambió? Le dimos la variable de confusión. Al comparar estudiantes 
con la misma motivación, el sesgo de selección desapareció. Eso es lo que significa «controlar» por una 
variable, y lo verás a fondo en el capítulo 5. 

Mira también el coeficiente de MOTIVACION: 2,03. Es casi justo lo que pusimos al crear los datos. 

Paso 7. Cuarto intento: el experimento 

¿Y si, en lugar de dejar que cada estudiante eligiera, sorteáramos las clases? Vamos a usar a los mismos 
estudiantes, con la misma motivación y la misma suerte en el examen. Lo único que cambia es quién decide: 
ahora decide una moneda. 
series clases_azar = @rbinom(1, 0.5) 
series nota_azar = 11 + 1*clases_azar + 2*motivacion + ruido 
nota_azar = @recode(nota_azar<0, 0, @recode(nota_azar>20, 20, nota_azar)) 
equation bal07.ls motivacion c clases_azar 
show bal07 

FIGURA 1.8   Salida de EViews: prueba de balance de la motivación entre los grupos sorteados. Fuente: elaboración propia 
con EViews 14 y datos simulados (cap01.prg). 

Antes de estimar la nota, revisamos el balance: ¿los dos grupos sorteados tienen la misma motivación? Para 
comprobarlo usamos el truco del paso 5 al revés. Una regresión de la motivación sobre la variable del sorteo 
compara los promedios de los dos grupos: la constante es el promedio del grupo sin clases y el coeficiente, la 
diferencia. 

La diferencia es de apenas −0,03, y su valor p es 0,27. No hay ninguna diferencia que merezca ese nombre. Si 
prefieres la prueba t clásica, la vista motivacion.testby clases_azar te da el mismo estadístico t, con el 
signo cambiado. 

El sorteo repartió la motivación de forma pareja sin que tuviéramos que medirla. Ahora sí, la regresión 
simple. 
equation eq08.ls nota_azar c clases_azar 
show eq08 

FIGURA 1.9   Salida de EViews: regresión simple de la nota sobre las clases sorteadas. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

Sin ningún control, el coeficiente es 0,87, con un error estándar de 0,07. Queda a menos de dos errores estándar 
del 1 verdadero: el intervalo de confianza al 95 % lo contiene. El azar hizo el trabajo que en el tercer intento hizo 
el control. ¿Por qué no sale 1,00 exacto? Porque un solo sorteo nunca reparte todo de forma perfecta; con otra 
semilla saldría un poco por encima o un poco por debajo. 

Fíjate en el R cuadrado: apenas 0,03. La regresión explica poco de la nota, porque la motivación y la suerte 
pesan mucho. Y aun así el efecto está bien medido. Un R cuadrado bajo no significa que el coeficiente esté mal. 
Imagina un programa que sube la nota un punto a cada estudiante: sería un gran programa, aunque explique poco 
de por qué unos sacan 8 y otros 18. 

Paso 8. La prueba del delito 

Volvamos a los datos originales, donde cada uno eligió. Comparemos la motivación de los dos grupos con la 
misma prueba de balance. 
equation bal09.ls motivacion c clases 
show bal09 

FIGURA 1.10   Salida de EViews: motivación promedio según la elección de clases particulares. Fuente: elaboración propia 
con EViews 14 y datos simulados (cap01.prg). 

Ahí está el sesgo de selección, a la vista. Los que no tomaron clases tenían, en promedio, una motivación de 
−0,54 (la constante). Los que sí, 1,07 más: alrededor de 0,53. El estadístico t es 44,42 y el valor p es 0,0000. 

Como cada unidad de motivación vale 2 puntos de nota, esa sola diferencia explica unos 2,14 de los 3,15 
puntos que vimos en el primer intento. Haz la cuenta: 1 punto de efecto real más 2,14 de sesgo. La ecuación de la 
trampa, con números. 

Paso 9. Controlar no es magia: la proxy ruidosa 

El tercer intento funcionó porque teníamos la motivación exacta de cada estudiante. En tu investigación no la vas 
a tener. Ninguna encuesta pregunta «¿cuán motivado estás, en una escala perfecta?». Lo que sí suele haber es 
algo parecido, por ejemplo las horas de estudio que el estudiante declara. A esa variable parecida se le llama 
proxy o variable sustituta. 

¿Basta con controlar por la proxy? Probemos. Vamos a crear unas horas de estudio que dependen de la 
motivación, pero con mucho ruido, como pasa con cualquier respuesta de encuesta. 
series horas = 6 + 2*motivacion + 3*nrnd 
horas = @recode(horas<0, 0, horas) 
group g10 motivacion horas 
show g10.cor 

FIGURA 1.11   Salida de EViews: correlación entre la motivación y las horas de estudio. Fuente: elaboración propia con 
EViews 14 y datos simulados (cap01.prg). 

Las horas de estudio se parecen a la motivación: la correlación es 0,54. ¿Será suficiente? 
equation eq11.ls nota c clases horas 
show eq11 

FIGURA 1.12   Salida de EViews: regresión de la nota sobre las clases con las horas de estudio como proxy de la 
motivación. Fuente: elaboración propia con EViews 14 y datos simulados (cap01.prg). 

No lo es. El coeficiente de las clases apenas bajó de 3,15 a 2,64. Seguimos muy lejos del 1 verdadero, y la tabla 
sigue luciendo impecable, con su valor p de 0,0000. 

La lección es incómoda, pero vale oro: controlar por una medida imperfecta del confusor solo corrige 
una parte del sesgo. Cuanto peor mide la proxy al confusor, más sesgo queda. Por eso un buen econometrista no 
se conforma con «ya controlé por todo lo que tenía». Se pregunta qué tan bien mide cada control lo que de 
verdad le preocupa y, cuando la respuesta es «más o menos», busca una estrategia distinta. Esas estrategias 
llenan la última parte del libro. 

Pongamos todos los intentos lado a lado. 

TABLA 1.1   Cinco formas de estimar el efecto de las clases particulares (efecto real: 1 punto) 

Método Efecto estimado ¿Acierta? 
Comparar promedios 3,15 No: triplica el efecto 
Regresión simple 3,15 No: es lo mismo con más decimales 
Regresión con el confusor 0,97 Sí, si mides el confusor 
Regresión con una proxy del confusor 2,64 No: corrige solo una parte 
Experimento (sorteo) 0,87 Sí, dentro de su margen de error, sin 

medir nada más 
Fuente: elaboración propia con EViews 14 y datos simulados (cap01.prg); cifras de las figuras 1.4, 1.5, 1.7, 
1.12 y 1.9. 

1.8   El error que comete casi todo tesista 

Abre el repositorio de tesis de cualquier universidad y busca títulos que empiecen con «Influencia de…» o 
«Impacto de…». Vas a encontrar muchos. Muchos se apoyan en una encuesta aplicada una sola vez, a un solo 
grupo, sin sorteo y sin ninguna estrategia para enfrentar el sesgo de selección. Y concluyen que X «influye 
significativamente» en Y porque el valor p salió menor que 0,05. 

El error no está en el cálculo. Está en la palabra. «Significativo» quiere decir que la relación probablemente 
no es producto del azar: descarta la cuarta explicación de nuestra lista. No dice nada de las otras tres. Un 
coeficiente puede ser significativo y estar contaminado por causalidad inversa o por un confusor, como acabas de 
ver con el 3,15. 

¿Cómo lo evitas? Con dos hábitos. 

Primero, elige el verbo según tu diseño. Si tu estrategia no elimina el sesgo de selección, escribe «se asocia 
con», «se relaciona con», «se observa una diferencia de». Reserva «efecto», «impacto» e «influye» para cuando 
tengas un experimento o uno de los métodos de los capítulos 12 a 14. Tu jurado lo va a notar. Y si no lo nota, lo 
notará quien lea tu tesis después. 

Segundo, antes de correr cualquier regresión, escribe tu lista de confusores. Pregúntate: ¿qué cosa podría 
mover al mismo tiempo mi variable explicativa y mi resultado? Si la tienes en los datos, contrólala. Si no la 
tienes, dilo en tus limitaciones y explica hacia dónde sesgaría tu resultado. Esa sola página hace más por tu 
investigación que diez pruebas de normalidad. 

Y su hermano gemelo: controlar por todo 

Hay un error opuesto, igual de frecuente. El tesista que ya aprendió que debe controlar mete en la regresión todas 
las variables que encuentra en el workfile, «por si acaso». Suena prudente, pero no lo es. 

Algunas variables no deben controlarse nunca, porque son consecuencia del tratamiento que estudias. 
Supongamos que quieres medir cuánto sube el ingreso de un agricultor por recibir un crédito y controlas por 
cuántas hectáreas sembró esa campaña. Parece razonable, hasta que te preguntas para qué sirvió el crédito: en 
buena parte, para sembrar más. 

Al comparar solo a agricultores que sembraron lo mismo, borras justo uno de los caminos por los que el 
crédito sube el ingreso, y además comparas grupos que ya no son comparables. Angrist y Pischke llaman a estas 
variables malos controles (Angrist & Pischke, 2009, pp. 64-68). 

La regla práctica cabe en una línea: controla por lo que ocurrió antes del tratamiento, no por lo que el 
tratamiento provocó. Vuelve a tu dibujo de flechas. Si una variable recibe una flecha desde tu tratamiento, 
déjala fuera. 

1.9   El error que cometen los investigadores en la redacción de artículos 

El error típico aparece en el resumen: «Los resultados muestran que la educación tiene un impacto positivo en el 
ingreso». Tres páginas después, en las limitaciones, el autor admite que sus datos son observacionales. El revisor 
lee primero el resumen y luego busca el diseño, y cuando no lo encuentra, el artículo vuelve con la palabra 
«causalidad» subrayada. 

Primero, los verbos cambian a mitad del texto. El método dice «asociación», los resultados dicen «efecto» 
y las conclusiones recomiendan una política como si el efecto estuviera probado. Un artículo tiene que usar el 
mismo verbo desde el título hasta la última línea. 

Segundo, la limitación no es un conjuro. Escribir al final «no se puede afirmar causalidad» no corrige un 
resumen que la afirma. Si tu diseño no elimina el sesgo de selección, dilo en la introducción, donde el lector 
decide si sigue leyendo. 

Cómo se escribe bien: «Quienes tomaron clases particulares obtienen 3,15 puntos más de nota, pero esa 
diferencia cae a 0,97 al comparar estudiantes con la misma motivación. Como los datos son observacionales, 
interpretamos la relación como una asociación y discutimos qué confusores podrían explicarla». 

RESUMEN Y CONCLUSIONES 

1.  Una correlación puede venir de causalidad, causalidad inversa, un confusor o el azar. El software no 
distingue entre las cuatro. 

2.  El efecto causal compara dos resultados potenciales de la misma persona, y uno de ellos nunca se observa. 

3.  Diferencia observada = efecto causal + sesgo de selección. Si los grupos no eran comparables, la 
diferencia te engaña. 

4.  Una regresión simple es una comparación de promedios. Un valor p pequeño no corrige el sesgo. 

5.  Controlar ayuda si mides bien el confusor; con una proxy imperfecta queda sesgo, y controlar por 
consecuencias del tratamiento lo empeora. 

6.  El sorteo elimina el sesgo de selección. Cuando no puedes sortear, la econometría te da herramientas para 
acercarte, cada una con su supuesto. 

En el capítulo 2 dejamos la simulación y trabajamos con datos reales: vas a descargar la ENAHO y los 
indicadores del Banco Mundial, y a dejarlos listos en EViews en una tarde. 

EJERCICIOS RESUELTOS 

Los tres están al final de cap01.prg, bajo el rótulo EJERCICIOS RESUELTOS. Córrelos después de los pasos 1 a 
9 y en este orden, porque usan el workfile que ya tienes abierto. Intenta predecir el resultado antes de correr cada 
uno. 

Ejercicio 1. ¿Y si la motivación no decidiera quién toma clases? 

Supongamos que los estudiantes siguen eligiendo, pero ahora lo que decide es la distancia entre su casa y la 
academia, de 0 a 10 kilómetros. Cuanto más lejos viven, menos probable es que paguen clases. La distancia no 
tiene nada que ver con la nota ni con la motivación. ¿Cuánto dará ahora la comparación simple de promedios? 
series distancia = 10*rnd 
series clases1 = @rbinom(1, @logit(2 - 0.4*distancia)) 
series nota1 = 11 + 1*clases1 + 2*motivacion + ruido 
nota1 = @recode(nota1<0, 0, @recode(nota1>20, 20, nota1)) 
show nota1.statby clases1 

FIGURA 1.13   Salida de EViews: nota promedio cuando la elección depende solo de la distancia. Fuente: elaboración propia 
con EViews 14 y datos simulados (cap01.prg). 

show motivacion.statby clases1 

FIGURA 1.14   Salida de EViews: motivación promedio cuando la elección depende solo de la distancia. Fuente: elaboración 
propia con EViews 14 y datos simulados (cap01.prg). 

Solución. La función rnd sortea un número entre 0 y 1, así que 10*rnd reparte la distancia al azar entre 0 y 10 
kilómetros. Los que tomaron clases promedian 11,96 y los que no, 10,98: una diferencia de 0,98, casi el 1 
verdadero. Y eso que la motivación sigue subiendo la nota 2 puntos por unidad. La segunda tabla explica por 
qué: la motivación promedio es −0,010 en un grupo y −0,009 en el otro, prácticamente lo mismo. 

Nadie lanzó una moneda y, aun así, la comparación acierta. Un confusor solo sesga tu comparación si mueve 
las dos cosas: el tratamiento y el resultado. Si lo que decide la elección no toca la nota, deja de ser un problema. 
Ojo: en tu investigación tendrías que defender ese supuesto con argumentos, porque los datos no pueden 
probarlo. Pensemos en la distancia real: ¿no viven más cerca de las academias las familias de barrios con más 
ingresos? Si es así, la distancia ya no sería inocente. 

Ejercicio 2. Pon número al sesgo de selección. 

Con los datos originales, calcula el sesgo de selección de dos formas. La primera, como la diferencia 
observada menos el efecto real. La segunda, como la diferencia de motivación entre grupos multiplicada por su 
efecto en la nota (2 puntos). 

equation e1.ls nota c clases 
scalar sesgo_a = e1.@coefs(2) - 1 
equation e2.ls motivacion c clases 
scalar sesgo_b = 2*e2.@coefs(2) 
table(3,2) tsesgo 
tsesgo.setwidth(1) 24 
tsesgo.setwidth(2) 12 
tsesgo(1,1) = "Forma de calcularlo" 
tsesgo(1,2) = "Sesgo" 
tsesgo(2,1) = "Observado - real" 
tsesgo(2,2) = sesgo_a 
tsesgo(3,1) = "Motivacion x 2" 
tsesgo(3,2) = sesgo_b 
show tsesgo 

 
FIGURA 1.15   Salida de EViews: el sesgo de selección calculado de dos formas. Fuente: elaboración propia con EViews 14 
y datos simulados (cap01.prg). 

Solución. Aquí aparecen tres herramientas nuevas. Una ecuación estimada con equation sin show se calcula en 
silencio: queda guardada en el workfile, pero no abre ventana. e1.@coefs(2) es el segundo coeficiente de esa 
ecuación (el primero es la constante). Y scalar lo guarda en un número con nombre. Para verlos juntos 
armamos una table de 3 filas y 2 columnas, le damos ancho a cada columna con setwidth y llenamos cada 
celda con tsesgo(fila, columna). 

Las dos cuentas casi coinciden: 2,150 y 2,145. Es la ecuación de la trampa comprobada con números: 3,15 
observados = 1 de efecto + unos 2,15 de sesgo. La pequeña diferencia entre ambas cuentas viene del ruido y del 
recorte de la nota entre 0 y 20. En tus datos reales nunca podrás hacer esta cuenta, porque no conoces el efecto 
real. Por eso conviene entrenar la intuición aquí, donde sí lo conoces. 

Ejercicio 3. Un mal control en acción. 

Ahora las clases se sortean, y además hacen que el estudiante estudie 3 horas más por semana. Cada hora 
sube la nota 0,2 puntos, y las clases suman 0,4 puntos directos. ¿Qué pasa si «controlas» por las horas? 
series c3 = @rbinom(1, 0.5) 
series horas3 = 5 + 3*c3 + 2*nrnd 
series nota3 = 11 + 0.4*c3 + 0.2*horas3 + 1.5*nrnd 
nota3 = @recode(nota3<0, 0, @recode(nota3>20, 20, nota3)) 
equation eq15.ls nota3 c c3 
show eq15 

FIGURA 1.16   Salida de EViews: efecto de las clases sorteadas sin controlar por las horas de estudio. Fuente: elaboración 
propia con EViews 14 y datos simulados (cap01.prg). 

equation eq16.ls nota3 c c3 horas3 
show eq16 

FIGURA 1.17   Salida de EViews: efecto de las clases sorteadas controlando por las horas de estudio (mal control). Fuente: 
elaboración propia con EViews 14 y datos simulados (cap01.prg). 

Solución. El efecto total de las clases es 1 punto: 0,4 directos más 0,6 que llegan porque el estudiante estudia más 
(3 horas × 0,2). La regresión sin controles lo estima bien, en 0,99, con un error estándar de 0,04. Al controlar por 
las horas, el coeficiente cae a 0,37. Le quitaste a las clases el camino por el que producen buena parte de su 
efecto. Las horas recibían una flecha desde el tratamiento, así que eran un mal control. 

Para tu tesis o artículo 

1.  Escribe tu pregunta como un «¿qué habría pasado si…?». Por ejemplo: «¿Cuánto habría ganado este 
agricultor sin el crédito?». Si no puedes escribirla así, quizás tu investigación es descriptiva. Eso también está 
bien: solo cambia los verbos. 

2.  Haz tu lista de tres confusores. Para cada uno, anota si está en tus datos y hacia dónde movería tu 
resultado si lo omites. 

3.  Revisa el título de tu tesis o de tu artículo. Si dice «influencia», «efecto» o «impacto», pregúntate qué 
parte de tu diseño elimina el sesgo de selección. 

Lo que te va a preguntar el jurado: «¿Cómo sabe que esa relación es causal y no una correlación?». 
Después de este capítulo tienes cómo responder: nombra el confusor que más te preocupa y explica qué hiciste 
con él. 

Si es un artículo: En la revista, la pregunta del jurado la hace el revisor, y la hace antes: si el título o el 
resumen dicen «efecto» o «impacto», buscará en los primeros párrafos de la introducción qué diseño elimina el 

sesgo de selección. Si tu diseño no lo elimina, cambia desde el título los verbos a «se asocia con»: es más barato 
que un rechazo. 

